
Program
#include <ESP8266WiFi.h>
#include <ESP8266WebServer.h>
// === Wi-Fi точка доступу ===
const char* ssid = "CivicShield_AP";
const char* password = "12345678"; // можна змінити
ESP8266WebServer server(80);
const int buttonPin = D2;
const int ledPin = D5;
bool sosActive = false; // початковий стан
// --- Головна сторінка ---
void handleRoot() {
 String html = "<html><head><meta charset='utf-8'>";
 html += "<style>";
 html += "body { font-family: Arial; text-align:center; margintop:40px; }";
 html += "button { font-size:20px; padding:10px 20px; border:none; border-radius:10px; cursor:pointer; }";
 html += ".on { background-color:red; color:white; }";
 html += ".off { background-color:green; color:white; }";
 html += "</style></head><body>";
 html += "<h2>CivicShield SOS Beacon</h2>";
 html += sosActive ? "<h3 style='color:red'> Світлов увімкнути!</h3>" : "<h3>Світло вимкнено!</h3>";
 html += "<form action='/toggle' method='POST'>";
 if (sosActive) {
 html += "<button class='off' type='submit'>Вимкнути світло</button>";
 } else {
 html += "<button class='on' type='submit'>Увімкнути світло</button>";
 }
 html += "</form>";
 html += "<p><small>ESP8266 Wi-Fi IP: " + WiFi.softAPIP().toString() + "</small></p>";
 html += "</body></html>";
 server.send(200, "text/html", html);
}
// --- Обробка натискання кнопки на сайті ---
void handleToggle() {
 sosActive = !sosActive;
 digitalWrite(ledPin, sosActive ? HIGH : LOW);
 Serial.println(sosActive ? "Світло активовано через веб!" : "Світло вимкнено через веб!");
 server.sendHeader("Location", "/");
 server.send(303); // перенаправлення назад на головну
}
void setup() {
 Serial.begin(115200);
 Serial.println();
 Serial.println("Запуск Wi-Fi точки доступу...");
 pinMode(buttonPin, INPUT_PULLUP);
 pinMode(ledPin, OUTPUT);
 digitalWrite(ledPin, LOW); // вимикаємо LED при старті
 WiFi.softAP(ssid, password);
 Serial.print("ESP Wi-Fi запущено. Підключайся до: ");
 Serial.println(ssid);
 Serial.print("IP адреса: ");
 Serial.println(WiFi.softAPIP());
 server.on("/", handleRoot);
 server.on("/toggle", HTTP_POST, handleToggle);
 server.begin();
 Serial.println("HTTP сервер запущено.");
}
void loop() {
 server.handleClient();
 static bool lastButtonState = HIGH;
 bool currentButtonState = digitalRead(buttonPin);
 // натискання фізичної кнопки
 if (currentButtonState == LOW && lastButtonState == HIGH) {
 sosActive = !sosActive;
 digitalWrite(ledPin, sosActive ? HIGH : LOW);
 Serial.println(sosActive ? "Світло активовано кнопкою!" : " Світло вимкнено кнопкою!");
 delay(300); // антидребезг
 }
 lastButtonState = currentButtonState;
}

